<commit_before>SUMMARY = "updatehub's device attributes scripts"
DEPENDS = "bind jq"
LICENSE = "MIT"
LIC_FILES_CHKSUM = "file://COPYING;md5=838c366f69b72c5df05c96dff79b35f2"
SRC_URI = "git://github.com/updatehub/device-attributes;protocol=https"
SRCREV = "cb588481493e33d773609aeabc9fe4f37e9dbeaf"

S = "${WORKDIR}/git"

inherit allarch

do_configure[noexec] = "1"
do_compile[noexec] = "1"

do_install() {
    install -Dm 0755 cpu_model ${D}/${datadir}/updatehub/device-attributes.d/cpu_model
    install -Dm 0755 kernel ${D}/${datadir}/updatehub/device-attributes.d/kernel
    install -Dm 0755 mem_total ${D}/${datadir}/updatehub/device-attributes.d/mem_total
    install -Dm 0755 ipinfo-io ${D}/${datadir}/updatehub/device-attributes.d/ipinfo-io
}

python split_plugins_packages () {
    device_attribute_dir = d.expand('${datadir}/updatehub/device-attributes.d/')
    do_split_packages(d, device_attribute_dir, '^(.*)', 'updatehub-device-attribute-%s',
                      'updatehub %s device attribute script', extra_depends='')
}

PACKAGES_DYNAMIC = "updatehub-device-attribute-*"
PACKAGESPLITFUNCS_prepend = " split_plugins_packages "

RDEPENDS_updatehub-device-attribute-ipinfo-io += "jq bind-utils"
<commit_msg>updatehub-device-attributes: Bump revision to c7ee915<commit_after>SUMMARY = "updatehub's device attributes scripts"
DEPENDS = "bind jq"
LICENSE = "MIT"
LIC_FILES_CHKSUM = "file://COPYING;md5=838c366f69b72c5df05c96dff79b35f2"
SRC_URI = "git://github.com/updatehub/device-attributes;protocol=https"
SRCREV = "c7ee915b0d304cec33a08cea7d3f184150c88dd1"

S = "${WORKDIR}/git"

inherit allarch

do_configure[noexec] = "1"
do_compile[noexec] = "1"

do_install() {
    install -Dm 0755 cpu_model ${D}/${datadir}/updatehub/device-attributes.d/cpu_model
    install -Dm 0755 kernel ${D}/${datadir}/updatehub/device-attributes.d/kernel
    install -Dm 0755 mem_total ${D}/${datadir}/updatehub/device-attributes.d/mem_total
    install -Dm 0755 ipinfo-io ${D}/${datadir}/updatehub/device-attributes.d/ipinfo-io
}

python split_plugins_packages () {
    device_attribute_dir = d.expand('${datadir}/updatehub/device-attributes.d/')
    do_split_packages(d, device_attribute_dir, '^(.*)', 'updatehub-device-attribute-%s',
                      'updatehub %s device attribute script', extra_depends='')
}

PACKAGES_DYNAMIC = "updatehub-device-attribute-*"
PACKAGESPLITFUNCS_prepend = " split_plugins_packages "

RDEPENDS_updatehub-device-attribute-ipinfo-io += "jq"

{'fraction_non_alphanumeric': 0.14152850788515972, 'fraction_numerical': 0.05135463000404367, 'mean_word_length': 5.231738035264484, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5625938', 'n_tokens_mistral': 997, 'n_tokens_neox': 909, 'n_words': 133}